
his newsletter is the home of Foundation SAS and brings 
you information about making the most of SAS 9 and 
Altair  SLC.  This bumper issue includes articles from 

this  year’s  August  and  November  issues,  which  have  been 
combined  due  to  holidays  and  illness,  after  my  retirement  in 
April 2025, delayed publication until now.  This newsletter issue 
includes  2  articles  from LeRoy Bessler  about  improving  time 
series  or  trend  graphs  and  vertical  bar  charts,  and  I  have 
contributed a Workflow tip from an upcoming book about Altair 
SLC.  My 2nd Editions of my “SAS Programming Experiences” 
and “Altair SLC” paperbacks are now available on Amazon and 
other resellers, but please read my article to get copies at the best 
available prices.  As usual I have publicised some SAS-related 
events for the next few months, and I have also added some more 
SAS formats,  options and functions that  you may or  may not 
have  come across  before,  but  you  might  find  interesting,  and 
even useful.

T

If  you  would  like  to  contribute  an  article,  to  re-visit  and 
improve  an  existing  article,  or  just  discuss  the  possibility  of 
doing  so,  please  feel  free  to  send  an  email  to  me  at 
views-uk@hollandnumerics.org.uk.

Philip R Holland (Newsletter Editor)

Membership of VIEWS is free and you can register for 
notification of all future VIEWS News issues by registering at 

blog.hollandnumerics.org.uk, and then emailing 
views-uk@hollandnumerics.org.uk, or using the blog site 

menu, to request access to the VIEWS UK forum. Members will 
receive an email notification whenever a new issue of the free 

quarterly VIEWS News is published on the web site, please 
inform us when you change your email address, otherwise your 

registration may be cancelled.

Ask The Consultant
This  part  of  VIEWS  News is  where  you  can  get  your 
technical questions answered.  Send your questions to the 
Editor.

What is  the simplest  way to present  a time 
series or trend in full?

I am going to show you coding that is somewhat complicated. 
To  put  it  in  a  fair  context,  let  me  first  show  you  the 
under-informative  EASY  alternative.   The  only  non-default 
coding was to size the TITLE to avoid a line break and wrapped 
text, and inclusion of the example’s self-evaluating FOOTNOTE. 
Of course, that’s really MY evaluation of the image.

I concede that this is NOT totally useless.  And I could spend 
time and space on progressive stepwise improvements, but let’s 
get to the best result.  More tick mark values would obviously 
help,  but  the  precise  Y  values  would  NEVER  be  actually 
knowable.   We don’t  really  NEED tick mark values  at  all,  to 
facilitate unwise guessing what the Y values MIGHT be.

Direct  annotation  of  the  Y  values  is  ALWAYS the  better 
solution, but the problem with any FULL annotation of a trend 
line is the possibility of collisions between the annotation and the 
line.  Collisions between annotation and marker (plot symbol) are 
less likely since the PROC SGPLOT annotation algorithm does 
its best to avoid that.  When I was working on examples of time 
series plots for my book  Visual Data Insights Using SAS ODS 
Graphics:  A  Guide  to  Communication-Effective  Data 
Visualization,  I  was  experimenting  with  ways  to  minimise 
collisions between annotation and line.   I  could have tried the 
minimum  line  width  and  a  very  faint  line  colour.   Instead  I 
decided  to  plot  the  markers,  but  make  the  line  colour  white. 
When I looked at that result, I realised that I DID NOT REALLY 
NEED the markers either.   That  is  what inspired the example 
below.

For these images,  because my ODS GRAPHICS statement 
includes no value for HEIGHT, the image height is 0.75 times 
the  value  of  WIDTH.   To  fit  more  than  one  year  of  months 
without label collisions at a readable point size for the text would 
require a wider image than this column width of 3.65 inches can 
accommodate.  I did verify that, for example, five years of this 
data  could  be  plotted  with  a  7pt  Arial  Bold  font  without 
collisions in a 13.33 inches BY 7.5 inches image, which is the 
size of a wide screen slide.   It  IS a tight fit,  and if the Close 
values  were  four  digits  (requiring  a  comma5.  format),  there 
would probably  be  collisions between some label  pairs.   That 
Full Annotation design would be best suited for ranges up to 3 or 
4 years.

This image, without the customary plot line and plot markers, 
IS everything that the viewer needs to know and see about the 
data.  The chain of annotations IS the trend.  It is a “virtual” trend 
line, but the viewer does not need special VR goggles to see it.

The plot includes the display of the Y=0 point, to reassure the 
viewer that the whole plot is really there and that the changing 
slope  of  the  invisible  line  is  a  fair  representation  of  the  time 
series behaviour.

Like  so  many  value-added  data  graphic  examples,  some 
pre-processing is required. 

First, to customise text characteristics in one fell swoop, let’s 
use a macro to create a custom ODS style that can control all of 
any  graph’s  text.   Overrides  by  using  the  text  characteristics 
controls for XAXIS, YAXIS, TITLE, and TEXT statements are 
nevertheless  possible.   This  macro and the  simple  addition of 
STYLE=  to  your  ODS  destination  statement  (here,  ODS 
LISTING) is always a coding time saver.  Your custom style is 
permanently saved in your SASUSER.TEMPLAT catalog. Run 
the macro once for all of the font sizes that you are likely to use, 
and for your preferred font “family”.  There is no benefit in NOT 
using  the  macro  default  Weight=Bold.   Bold  never  impairs 
readability, unless one is averse to more easily SEEING text. 

%MACRO AllGraphTextSetup(Size, Family=Arial,
                         Weight=Bold);
  %LET FamilyForStyle =
    %SYSFUNC(COMPRESS(&Family,' '));
  PROC TEMPLATE;
    DEFINE STYLE
      GraphFont&FamilyForStyle.&Size.pt&Weight;
      PARENT=STYLES.LISTING;
      CLASS GRAPHFONTS /
'GRAPHVALUEFONT' = ("&Family",&Size.pt,&Weight)
'GRAPHLABELFONT' = ("&Family",&Size.pt,&Weight)
'GRAPHDATAFONT'  = ("&Family",&Size.pt,&Weight)
'GRAPHTITLEFONT' = ("&Family",&Size.pt,&Weight)
'GRAPHFOOTNOTEFONT' =
                   ("&Family",&Size.pt,&Weight);
    END;
  RUN;
%MEND AllGraphTextSetup;

%AllGraphTextSetup(8);

/* Find this message in your SAS log:
NOTE: STYLE 'GraphFontArial8ptBold' has been 
saved to: SASUSER.TEMPLAT */

DATA WORK.IBM1998Close;
  RETAIN MaxClose 0;
  SET SASHELP.STOCKS END=LastOne;
  WHERE YEAR(Date) EQ 1998 AND Stock EQ 'IBM';
  OUTPUT;
  MaxClose = MAX(MaxClose, Close);
  IF LastOne;
  CALL SYMPUT('MaxClose_1998', MaxClose);
RUN;

DATA work.IBM1998CloseAndDataLabels;
  KEEP Date Close DataLabelForYandX;
  LENGTH DataLabelForYandX $7;
  SET work.IBM1998Close;
  DataLabelForYandX = PUT(Close, 3.) || ','
       || PUT(Date, MONNAME3.);
RUN;

ODS RESULTS OFF;
ODS _ALL_ CLOSE;
ODS LISTING STYLE=GraphFontArial8ptBold
  GPATH="C:\TEMP";
ODS GRAPHICS ON / RESET=ALL
  SCALE=OFF WIDTH=3.65in
  IMAGENAME="IBMMonthlyFirstTradingClose1998";

TITLE1 JUSTIFY=CENTER
'1998 Monthly First Trading Day Close Price of 
IBM Shares';
TITLE2 COLOR=WHITE 'white space';

PROC SGPLOT DATA=WORK.IBM1998CloseAndDataLabels
            NOBORDER;
  WHERE YEAR(Date) EQ 1998;
  TEXT X=Date Y=Close TEXT=DataLabelForYandX 
    / CONTRIBUTEOFFSETS=(XMIN XMAX YMAX)
/* The purpose of the offsets is to protect 
displayed values that are near the ends of the X 
axis from being clipped. YMIN would needlessly 
raise the 0 value, and would vertically compress 
the plot. */
    SPLITCHAR=',' SPLITJUSTIFY=CENTER
    SPLITPOLICY=SPLITALWAYS;
/* If SPLITPOLICY=SPLITALWAYS is omitted, the 
TEXT values are presented concatenated,    
rather than stacked, with collisions for many 
pairs of labels. */
  YAXIS DISPLAY=(NOLINE NOTICKS NOLABEL)
    VALUES=(0 &MaxClose_1998.) 
    VALUESDISPLAY=("0" " ");
  XAXIS DISPLAY=NONE;
  FORMAT Date MONNAME3. CLOSE 3.;
RUN;

ODS LISTING CLOSE;
TITLE; FOOTNOTE;

The data-specific code above could be revised and adapted 
for a general-purpose macro.  If you take that on and have any 
problem or question, an email to me will get a reply.  The macro 
itself would be more complicated, but once it has been written, it 
would be a time saver.  If you do it,  please share it  here in a 
future issue of VIEWS News.

LeRoy Bessler, Bessler Consulting and Research

What is a satisfactory alternative vertical bar 
chart?

I have shown you a time series plot with no line and no markers. 
Now, analogously I will show how a bar chart with no bars can 
be a satisfactory alternative to a bar chart WITH bars.

Let’s look at a reasonable minimum best effort to improve 
this from the standpoint of knowing approximately what the sales 
values MIGHT be.  In any case, two of my universally applicable 
rules for graphs of categorical data are:

(A) Rank the response (high to low [when higher is better], 
or  low  to  high  [when  lower  is  better]).   Show  them  what’s 
important.

(B) List the Grand Total response in TITLE1 or TITLE2. 
Put each response value in context.

Since a grid is only an aid to estimating better, let’s annotate 
the bars with the response values instead.

The data labels at the top of the bars are rendered with the 
Arial Narrow font with the same font size as the remaining Arial 
text.

Those data labels, if Arial, would be drawn vertically up from 
the bar tops.  I do not read with my head tipped sideways at 90 
degrees.

Data labels at the top of bars is the default position.  An easy 
courtesy to provide your viewer is to display them at the bottom 
of the bars.  Then category and response value are in immediate 
association.   One looks  at  the  bars  and their  heights  only  for 
visual comparison, not for precise knowledge.

Besides Grand Total response in a title or subtitle, as context, 
each category’s response as Percent of Grand Total is a sharper 
indicator  of  context.   A  pair  of  X  axis  tables  provides  both 
response and percent.

Having  all  three  of  the  key  data  elements  for  a  chart  of 
categorical  data  in  immediate  association  as  shown  above 
inspired me to  use them stand-alone as a surrogate for the bar 
ends.  Here is my vertical bar chart without bars.

The  bar  ends,  if  drawn,  would  be  halfway  between  the 
response value and the first  (or only) line of the category and 
percent values.  Here, the percent value in the category-percent 
string  drops  into  the  body  of  the  bar.   If  the  width  of  this 
newsletter column were wide enough, a much better version of 
this chart would display the un-split value of the category and 
percent  on  one  line.   The  undrawn  bar  end  would  be  in  the 
middle of only two lines of information.

This image, without the customary bars, IS everything that 
the viewer needs to know and see about the data,  and it  does 
show how the categories compare.

One  of  the  most  important  of  the  design  principles  for 
communication-effective data visualisation is:

Deliver Visual AND Precise Numbers: Visual for quick easy 
inference plus Precise Numbers for reliable accurate inference.

Visual - For Impression:

1. Immediate

2. Easy

3. Memorable (“Sticky”)

Precise Numbers - For Knowledge:

1. Clear

2. Certain

3. Correct (“The Right Answer”)

In  the  bar  chart  without  bars,  the  precise  numbers  (and 
category value) ARE the visual.

The plot includes the display of the Y=0 point, to reassure the 
viewer  that  the  whole  chart  is  really  there  and  it  is  a  fair 
comparison of the categories.

Below is the code for the chart with two X axis tables and for 
the bar chart without bars.  Use whichever solution you prefer. 
Both  deliver  everything  needed  and  are  equally 
communication-effective.

PROC SUMMARY DATA=sashelp.shoes;
  CLASS Product;
  VAR Sales;
  OUTPUT OUT=work.AllTotals SUM=Sales;
RUN;

DATA work.ToChart;
  LENGTH ProductAndPercent $21;
  RETAIN GrandTotal 0 MaxSales 0;
  SET work.AllTotals END=LastOne;
  IF _TYPE_ EQ 0 THEN DO;
    GrandTotal = Sales;
    CALL SYMPUT('GrandTotal',
      PUT(GrandTotal, DOLLAR11.));
  END;
  ELSE DO;
    MaxSales = MAX(Sales, MaxSales);
    IF LastOne
      THEN CALL SYMPUT('MaxSales', MaxSales);
    ProductAndPercent = TRIM(LEFT(Product))
      || ' ' || 
      PUT(((Sales / GrandTotal) * 100), 4.1)
      || '%';
    OUTPUT;
  END;
RUN;

ODS RESULTS OFF;
ODS _ALL_ CLOSE;
ODS LISTING GPATH="C:\temp" DPI=300;
ODS GRAPHICS ON / 
  RESET=ALL SCALE=OFF WIDTH=3.65IN
  IMAGENAME="RankedVbarChartWithTwoXaxisTables";

TITLE1 JUSTIFY=CENTER HEIGHT=6PT
  "Shoe Sales by Product, Ordered By $";
TITLE2 JUSTIFY=CENTER HEIGHT=6PT
  "Grand Total Sales = &GrandTotal.";

PROC SGPLOT DATA=sashelp.shoes NOBORDER;
  VBAR Product / RESPONSE=Sales
    CATEGORYORDER=RESPDESC
    DISPLAYBASELINE=OFF NOOUTLINE;
  XAXISTABLE Sales / STAT=SUM
    VALUEATTRS=(FAMILY='Arial Narrow'
                SIZE=6PT WEIGHT=BOLD)
    POSITION=BOTTOM /* below the bars */
    LOCATION=INSIDE /* above axis values */
    NOLABEL; 

  XAXISTABLE Sales / STAT=PERCENT
    VALUEATTRS=(FAMILY='Arial Narrow'
                SIZE=6PT WEIGHT=BOLD) 
    POSITION=BOTTOM /* below the bars */
    LOCATION=INSIDE /* above axis values AND
      below any previously defined X axis
      table */
    NOLABEL; 
  YAXIS DISPLAY=NONE;
  XAXIS DISPLAY=(NOLINE NOTICKS NOLABEL)
    VALUEATTRS=(SIZE=6PT WEIGHT=BOLD)
    FITPOLICY=SPLITALWAYS SPLITCHAR=' ';
RUN;

ODS LISTING CLOSE;

PROC SORT DATA=work.ToChart 
          OUT=work.ByDescendingSales;
  BY DESCENDING Sales;
RUN;

ODS RESULTS OFF;
ODS _ALL_ CLOSE;
ODS LISTING GPATH="C:\temp" DPI=300;
ODS GRAPHICS ON / 
  RESET=ALL SCALE=OFF WIDTH=3.65IN
  IMAGENAME="RankedVbarChartWithoutBars";

TITLE JUSTIFY=CENTER HEIGHT=6PT BOLD
  COLOR=BLACK 'Ranked ' COLOR=Blue 'Shoe Sales'
  COLOR=BLACK ' and Percent Share By Product';
TITLE2 JUSTIFY=CENTER HEIGHT=6PT BOLD
  "Grand Total = &GrandTotal.";

PROC SGPLOT DATA=work.ByDescendingSales
            NOBORDER NOAUTOLEGEND;
  TEXT X=Product Y=Sales TEXT=Sales /
    POSITION=TOP
    TEXTATTRS=(COLOR=Blue FAMILY='Arial Narrow'
               SIZE=6PT WEIGHT=BOLD);
  TEXT X=Product Y=Sales TEXT=ProductAndPercent/
    POSITION=BOTTOM
    TEXTATTRS=(COLOR=BLACK FAMILY='Arial Narrow'
               SIZE=6PT WEIGHT=BOLD)
    SPLITPOLICY=SPLITALWAYS SPLITWIDTH=7
    SPLITCHAR=' ' SPLITJUSTIFY=CENTER;
  YAXIS DISPLAY=(NOLINE NOTICKS NOLABEL)
    VALUEATTRS=(SIZE=6PT WEIGHT=BOLD)
    VALUES=(0 &MaxSales.)
    VALUESDISPLAY=('0' ' ');
  XAXIS DISPLAY=NONE;
RUN;

ODS LISTING CLOSE;

To recap, my rules for bar charts are:

• Order the bars (real or virtual) by size, i.e. response value.

• Provide Percentage of Grand Total Response, not only the 
response value.

• Include the Grand Total in the chart title or subtitle.

LeRoy Bessler, Bessler Consulting and Research

Did You Know?
If you have a useful hint or tip, please send it to the Editor, 
and share it with the VIEWS membership.

Altair  SLC  Workflows  include  functionality 
similar to SAS Viya

[Altair SLC 2025] Altair  SLC  can  read  and  write  both 
SAS7BDAT and WPD data sets, and uses WPD data sets in its 
WORK library and in Workflows.  SAS software can only read 
and write SAS7BDAT data sets.

The following Workflow functionality is included in the free 
Personal Edition of Altair SLC from version 2025.

Creating a Neural Network

The following Workflow example uses the sashelp9.iris data 
set used in the SAS Viya neural network example.  However, the 
intermediate and final files are specific to Altair SLC, and are not 
readable by SAS software.

SAS Language

Note  that  the  sashelp9 library  reference  points  to  a  SAS 
Software installation.  This node copies the SAS data set into the 
expected  temporary  WPD  data  set,  which  is  referred  to  by 
&output_1.  There is also an iris.csv file in the installation file 
doc/en/sample-en.zip file that contains the same information, as 
the data is in the public domain.

Partition

The Partition node splits the input data set into 3 parts.

Multilayer Processing (MLP)

Multilayer processing is used in the SAS Viya neural network 
example.  No test dataset role needs to be specified, because the 
score_iris is one of the input datasets to the Score node.

In the SAS Viya example the target variable is Species, and 
the Interval Variables are SepalLength, SepalWidth, PetalLength 
and PetalWidth.

While not equivalent to the SAS Viya example, the Optimiser 
settings were selected here for the Workflow example.

Likewise, the Regularisers settings were selected here for the 
Workflow example.

Finally, the Stopping Criteria were selected to ensure that the 
neural network finished its analysis.

The results are shown below:

Score

The location of the output dataset is specified in the Score 
node.

The results look like this:

The  _INFO_  variable  has  been  truncated,  because  it  has 
assumed its length from the first value seen, which is 6 characters 
long: “Setosa”.  The scoring code can be improved, as follows:

This example is taken from a draft copy of “Advanced Altair SLC”
by Philip R Holland and Pablo Orosco Joerger

due to be published in 2026.

Philip R Holland (Newsletter Editor)

Formats, Options, and Functions
This  section  is  devoted  to  the  description  of  useful,  or 
unusual,  SAS Formats,  Options,  and Functions.   Please 
read the SAS Software and/or Altair SLC documentation to 
find out more about them.

[SAS and Altair SLC] The  numeric  formats  EUROw.d  and 
EUROXw.d display values in a similar way to  DOLLARw.d 
and  DOLLARXw.d,  respectively, but  with  a  prefix  of  “€” 
instead of “$”.  EUROw.d uses “,” for a thousands separator and 
“.”  for  a  decimal  separator,  while  EUROXw.d uses  “.”  for  a 
thousands  separator  and  “,”  for  a  decimal  separator,  as  is 
commonly used in Continental Europe.  There are also informats 
with the same names that  can be used to  read character  Euro 
values into numeric values.

[SAS and Altair SLC] The  EMAILHOST= system  option 
specifies  one  or  more  Simple  Mail  Transfer  Protocol  (SMTP) 
server domain names for your site to be used for sending emails 
directly from a program.

[SAS and Altair SLC] The CALL SOUND() routine, which 
is only available on Windows platforms, generates a sound with a 
specific frequency and duration.

Good examples for CALL SOUND can be found in “I’m Going to Pop 
Some Tags: Using SAS® CALL SOUND Routines” by Richard Dirmyer 

at www.lexjansen.com/nesug/nesug13/33_Final_Paper.pdf

Book Releases

SAS  Programming  Experiences  and  Altair 
SLC  paperbacks  are  now  on  Amazon  and 
other reseller sites

The original editions only included details of the 2024 versions 
of Altair SLC, but with the release of 2025 versions of Altair 
SLC,  it  seemed  right  to  revise  my  SAS  Programming 
Experiences and Altair SLC paperbacks and ebooks to include 
the new features of Altair SLC 2025.

Some of the windows in Altair  Analytics Workbench have 
changed  significantly  between  2024  and  2025,  but  the  major 
difference you will see is that the 2nd Editions are now available 
on Amazon, Barnes & Noble and Ingram, etc..  However, please 
do  not  buy  these  editions  there,  as  the  resellers  inflate  their 
prices,  but  wait  for  discounted  prices  for  the  paperbacks  on 
Lulu.com (which  should  be  in  every  “odd”  month:  Jan,  Mar, 
May,  Jul,  Sep  and  Nov),  or  download  the  equivalent  ebooks 
instead  from  https://hollandnumerics.org.uk/book-table,  which  is 
where the lowest prices for paperbacks and ebooks can always be 
found.

A new edition of the Altair  SLC book may be released in 
2026, which focuses on the 2026 version.  Anyone downloading 
the ebook from  https://hollandnumerics.org.uk/book-table will  be 
able to upgrade their previous download at no extra cost.

Also, the 3 books about Altair SLC I am co-authoring with 
Pablo Orosco Joerger are focusing on the current 2026 versions. 
These new books are planned for publication in early 2026 with 
the following provisional titles:

• Advanced Altair SLC
The target audience will be specifically SAS programmers 
using the free Personal Edition Altair SLC software, although 
all users of Altair SLC will benefit from this book.

• Altair SLC for Business Users
The target audience will be SAS programmers working in 
businesses using fully licensed Altair SLC software.

• Altair SLC for Companies
The target audience will be companies planning to migrate 
from SAS Software installations to Altair SLC installations.

Philip R Holland (Newsletter Editor)

In Brief
• Back issues of VIEWS News are available from the VIEWS UK 

forum at blog.hollandnumerics.org.uk.  Please register for free 
membership on the blog, and then request free access to the 
VIEWS UK forum.

• The SAS Professional Forum group on LinkedIn, which 
currently has over 41,300 members.  Everyone is welcome to 
request to join, but discussions here are restricted to 
SAS-related topics: www.linkedin.com/groups?gid=70702.

• The SAS Opportunities group on LinkedIn, which currently 
has over 23,300 members.  Everyone is welcome to request to 
join, but discussions here are also restricted to SAS-related 
topics: www.linkedin.com/groups?gid=805217.

• Altair SLC Personal Edition provides named users with free 
access to Altair SLC and Analytics Workbench software 
(formerly WPS) including a SAS language compiler with 
coding and workflow tools on unlimited data volumes, with 
interfaces to a wide range of databases, as well as integrated 
R and Python programming environments.  When the 
Personal Edition software is downloaded, it includes a licence 
for up to 12 months.  At the end of this term, users should 
download a new copy of the software to renew their licence 
and/or to upgrade to the latest available full version.  
Download the software from altairone.com/Marketplace.  No 
technical support service is offered with the Personal Edition, 
but assistance can be requested using the Community forums 
at community.altair.com.  The current General Access version 
of Altair SLC is 5.26.0 (2026).

• Do you use Microsoft Visual Studio Code Editor?

• You may be interested in a SAS-related extension called 
SASjs - DevOps for SAS.  You can develop SAS 
applications and run code on local SAS installations, SAS 
Viya, SAS 9 EBI, SASjs Server or local Altair SLC!  Just 
set up the target, and click the running man to execute 
your SAS code.  The SASjs Server could be your local 
SAS or Altair SLC installation.  SASjs is an open source 
project, and can be obtained from the Visual Studio Code 
Marketplace, where the current version is v2.6.0.

• The SAS Institute has also been developing its own 
extension called SAS.  You can develop SAS 
applications and run code on local SAS installations, SAS 
Viya, or SAS OnDemand for Academics!  Just set up the 
target, and click the running man to execute your SAS 
code.  The extension can be obtained from the Visual 
Studio Code Marketplace, where the current version is 
v1.18.0.

• Another Microsoft Visual Studio Code Editor extension has 
just been released called Altair SLC.  This extension 
currently only allows you to connect to local Altair SLC 
installations, even if, until now, you have only been able 
to use Altair Analytics Workbench to access Altair SLC.  
The current version in the Visual Studio Code 
Marketplace is v0.7.1 (pre-release).

Diary
Are you organising an event that would be of interest to the 
VIEWS readership?  Let us know as we are interested in all 
non-profit making events related to SAS.

March 2026
22-26 PhUSE US Connect, Austin, TX, USA

The  PhUSE  Community  allows  you  to  enhance  your 
learning and gain insights from our experts on today's most 
critical  industry  topics  through  a  variety  of  engaging 
presentations,  workshops,  and  networking  opportunities. 
Individuals from all areas of the industry will be gathered in 
one  place,  providing  an  opportunity  not  only  for  you  to 
advance your career but also to heighten the pharmaceutical 
industry’s  influence  and  findings.   Full  details  at 
www.phuse.eu.

May 2026
31 May-3 June PharmaSUG 2026, Boston, MA, USA

2.5 days of live presentations, hands-on training, e-posters 
and  panel  discussions.   Pre- and  post-conference  in-person 
training  seminars.   Networking  opportunities  during  the 
conference.  Full details at www.pharmasug.org.

November 2026
15-18 PhUSE EU Connect, Glasgow, UK

The  PhUSE  Community  allows  you  to  enhance  your 
learning and gain insights from our experts on today's most 
critical  industry  topics  through  a  variety  of  engaging 
presentations,  workshops,  and  networking  opportunities. 
Individuals from all areas of the industry will be gathered in 
one  place,  providing  an  opportunity  not  only  for  you  to 
advance your career but also to heighten the pharmaceutical 
industry’s  influence  and  findings.   Full  details  at 
www.phuse.eu.

LeRoy Bessler Email: besslerleroy2024@gmail.com

SAS UK Tel: +44-(0)1628-486933
Web: www.sas.com/uk

